

PEA Stata Code Manual

I. Introduction

The new PEA 3.0 framework provides an integrated analysis of countries' progress in advancing on the updated World Bank mission statement: *ending extreme poverty and boosting shared prosperity on a livable planet*. Key to the PEA revamp is a standardization effort to help to streamline the analysis.¹

This manual provides step-by-step guidance to World Bank staff for using a newly developed PEA Stata package, which automates the production of trends in poverty, shared prosperity and inequality, as well as their drivers and profiles. The package generates a full set of tables (`pea tables`) and figures (`pea figures`) to aide the core analytics in PEAs (shown in Annex 1: Full list of tables and figures). A smaller set of tables and figures are for inclusion in a standardized data annex (`pea core`) to ensure consistency and comparability across countries.

All outputs are automatically created by running the three commands, with minimum manual inputs needed from users. Parts of the codes can nevertheless be tailored to county contexts, for instance by specifying country-specific education or industry groups. The codes facilitate standardization of core analysis across countries and reproducibility of results. This guide explains how to install, set up, and execute the codes in Stata by following the below steps:

- Step 1: Installing and setting up the pea Stata package
- Step 2: Preparing and harmonizing survey data
- Step 3: Create additional variables
- Step 4: Run the pea commands

II. Installation and Setup

Step 1: Install the pea Stata Package

The PEA package can be downloaded from [Github](#) (click on “Code” on the top right and “Download ZIP”). Proceed to store the unzipped file in your folder of choice. The file location needs to be specified in the code later on (see Annex 2: Example code (adopath +)).²

The package includes the three main commands (`pea core`, `pea tables`, `pea figures`) which produce one Excel output file each. Additional ado-files allow each table and figure to be created separately. The whole package needs to be downloaded in order to produce the full output. This is because some of the ado-files run calculations and prepare and download data in the background. Ensure internet access is enabled, as additional data files are downloaded automatically from pip or datalibweb.

A data sub-folder needs to be extracted from the package and stored in the correct location on your device. A few additional datasets are required to run the full package, which are included in the pea

¹ See the [PEA website](#) for resources and tools for PEAs.

² An integration to be downloaded directly from Stata for automatic installation is planned.

package.^{3,4} The sub-folder within the package can be found here: “/pea/Stata/personal/**pea**”. Importantly, this **pea** sub-folder (including the Scorecard_Summary_Vision folder) needs to be moved into the personal folder of your Stata system directory (which can be found by entering “sysdir” in Stata): e.g. “c:/ado/personal/**pea**”.

Step 2: Access and Prepare Survey Data

The pea Stata package runs on survey data harmonized according to the Global Monitoring Database (GMD).⁵ Proceed to one of the following next steps.

(2a) Using pre-harmonized GMD data

If the survey data is already included in the GMD, access it in Stata using the `datalibweb` command (see Step 2 in Annex 2: Example code).⁶ Using `datalibweb` has the advantage of retrieving multiple years of data easily and consistently. You may proceed to Step 3.

(2b) If survey data needs to be harmonized and added to GMD

If the latest survey is not already harmonized in the GMD, country teams should work with the regional stats team to add the latest survey data into the database. If using multiple rounds of data for analysis, the new data needs to be appended to previous datasets (which can be accessed through `datalibweb`) before running the `pea` command, to include trends and changes in the tables and figures.

Step 3: Create additional variables and check data

Once all surveys have been GMD-harmonized (step 2), only the poverty lines and PPP-adjusted welfare aggregates need to be created manually. The specific variables that need to be created are PPP-adjusted and constant welfare aggregates, and the used poverty lines. In the example codes these variables are called: `welfareppp`, `pline215`, `pline365`, `pline685`, `natline` (and `natline2`—in case that there is a second

³ The following are included in the data folder: `CLASS.dta`, which includes country classifications, updated in July every year (<https://github.com/GPID-WB/Class/tree/master/OutputData>); `exposure_vulnerability_2021.dta`, which contains data needed to calculate exposure and vulnerability to climate-related hazards (accessed from the [reproducibility package](#) of the Poverty, Prosperity and Planet Report 2024); `UNESCO.dta`, which is used in the calculation for the multidimensional poverty measure (accessed through `datalibweb`); `Scorecard_Summary_Vision`, which contains excel files for each corporate scorecard indicator (accessed from World Bank [Scorecard website](#).)

⁴ The package also installs several additional Stata commands. The following Stata packages are installed automatically when running the `pea` code for the first time (and don't need to be installed separately): `apoverity`, `ineqdeco`, `svylorenz`, `fastgini`, `glcurve`, `alorenz`, `povdeco`, `fs`, `groupfunction`, `drdecomp`, `adecomp`, `pip`, `schemepack`, `colorpalette`, `geoplot`, `palettes`, `colrspace`, `moremata`. The command `sedecomposition` cannot be installed through Stata's SSC and is therefore included in the `pea` package (`pea/Stata/plus/s`). The `sedecomposition` package has been accessed from here: https://github.com/vavalomi/stata_tools/blob/master/sedecomposition/sedecomposition.ado

⁵ For more information, visit the [GMD website](#).

⁶ If this is your first time accessing `datalibweb`, visit <https://datalibweb2.worldbank.org/home> to create a token to access `datalibweb` through Stata. See here for [datalibweb guidelines](#), and the [D4G website](#) for more instructions.

national poverty line).⁷ Step 3 in Annex 2: Example code provides an example code to construct these additional variables. The PEA package also includes example do-files that can be used as templates to run the codes (“pea/Stata/Example”).

If users want tables to show standard-errors, the data needs to be ‘survey set’ in Stata, before running the code.⁸ In the example in Annex 2: Example code, the line “`svyset . .`” in Step 3 survey sets the data accordingly.

The `pea` commands have an option “`setting (GMD) ”`, which creates an additional set of variables. If the option `setting (GMD)` is not specified, variables “`married`” and “`nowork`” need to be created. Table 1 displays code to create these, and the other additional variables (such as “`welfppp`”).

Table 1 lists all variables that are needed to run the `pea` commands in Stata. It is critically important to check and confirm that all variables in the GMD-harmonized dataset take on the correct variable type, value labels, and only the allowed values. Specifically, all variables should be numeric and variable labels and value labels need to be defined. This is to ensure proper presentation and minimal post-editing in tables. For example, the value label of the subnational region variable should be in the format in which region names should be displayed in the table output (Annex 2: Example code shows an example in Step 3).

The type, value labels and values of all variables need to be consistent across all surveys. This should be carefully checked, in particular when appending new survey data to pre-harmonized GMD data. Household identifiers might be repeated between surveys over time, even though it is not panel data. In this case, please create year-specific identifiers.

Note that even when not all variables are in the dataset or correctly defined, the `pea` commands will still run and produce all tables and figures where these variables are not required.

Table 1: List of variables used by the PEA command

Description	Name	Values	Code
1) Variables that need to be created			
Welfare aggregate (divided by CPI and ICP, and 365 days)	welfppp	<i>integer</i>	gen welfppp = welfare/cpi2017/icp2017/365 (note: welfare should be in per-capita terms)
National poverty line (CU)	natline	<i>integer</i>	gen natline = X if year==
Second national poverty line (LCU)	natline2	<i>integer</i>	gen natline2 = X if year==
Poverty line: \$2.15 per day (PPP)	pline215	2.15	gen pline215 = 2.15
Poverty line: \$3.65 per day (PPP)	pline365	3.65	gen pline365 = 3.65

⁷ Importantly, verify that the correct deflators are used.

⁸ This is only relevant for Table 1, and Table A.1.

Poverty line: \$6.85 per day (PPP)	pline685	6.85	gen pline685 = 6.85
Subnational ID. Make sure the variable is numeric and harmonized over all years.	subnatvar	integer (labeled)	e.g. encode subnatid, gen(subnatvar)
2) Additional variables that are needed if setting(GMD) is not specified (see next section for more details on the code)			
Household head dummy	head	0=Not household head; 1=Household head	gen head = relationharm==1 if relationharm~.
Married dummy	married	0=not married; 1=married	gen married = marital==1 if marital~.
Not working dummy	nowork	0=working; 1=not working	gen nowork = lstatus==2 lstatus==3 if lstatus~.
3) Variables are needed to obtain the full set of outputs			
Household identifier	hhid	integer	
Individual identifier	pid	integer	
Year of survey	year	integer (4 digits)	
Poverty specific weights	weight_p	integer	
Welfare aggregate (LCU)	welfare	integer	
Relation to household head	relationharm	1=Household head; 2=Spouse; 3=Child; 4=Parents; 5=Other relative; 6=Non-relative	
Age of individual	age	integer	
Sex of individual	male	0=female; 1=male	
Household size	hsize	integer	
Marital status of individual	marital	1=married; 2=never married; 3=living together; 4=divorced/separated; 5=widowed	
Highest level of education completed (4 categories)	educat4	1=No education; 2=Primary (complete or incomplete); 3=Secondary (complete or incomplete);	

		4=Tertiary (complete or incomplete)	
Urban indicator	urban	1=urban; 2=rural	
Currently in school dummy	school	0=not currently in school; 1=currently in school	
Labor force Status	lstatus	1=employed ; 2=unemployed; 3=not in labor force	
Sector/industry of employment (4 categories) [NOTE: For some countries this is industrycat4_year]	industrycat4	1=Agriculture; 2=Industry; 3=Services; 4=Other	
Employment status	empstat	1=Paid employee; 2=Non-paid employee; 3=Employer; 4=Self-employed	
Improved water access	imp_wat_rec	0=No; 1=Yes	
Improved sanitation facility	imp_san_rec	0=No; 1=Yes	
Access to electricity	electricity	0=No; 1=Yes	
Ownership of a television	tv	0=No; 1=Yes	
Ownership of a car	car	0=No; 1=Yes	
Ownership of a cellphone	cellphone	0=No; 1=Yes	
Ownership of a computer	computer	0=No; 1=Yes	
Ownership of a fridge	fridge	0=No; 1=Yes	
Comparability between survey rounds	Comparability	<i>Integer</i>	

Step 4: Run pea commands

Once the pea package, data and additional variables have been prepared, users can follow the example in Annex 2: Example code under Step 4 to run the commands. The PEA package includes example do-files

that can be used as templates to run the codes (“pea/Stata/Example”). Note that the prepared data and the adopath to the pea package need to be called every time before a pea code is run (as in the example).

The syntax of the three main programs inside the `pea` code is outlined below. Stata help files are available for all commands (refer to Annex 1: Full list of tables and figures, for a list of all help files and table and figure numbers). Each table or figure can also be produced separately, using the table/figure specific command (see Step 4 in Annex 2: Example code). All tables and figures include standardized tables notes which can be customized as needed. The core appendix tables are a subset of the full list tables. The following codes produce the core appendix tables and figures:

- Table A.1: `pea table1`
- Table A.2: `pea tableA2`
- Table A.3: `pea table10`
- Table A.4a: `pea table4`
- Table A.4b: `pea table4`
- Figure A.1: `pea figure3b`
- Figure A.2: `pea figure4`

Table 2 provides details on the main options that need to be specified to produce output. Option `setting(GMD)`, when specified, automatically creates all the input variables needed to produce the entire core analytics output. Furthermore, when specified, a number of additional options do not need to be specified. These options are listed in Table 3. The `setting(GMD)` option works only in conjunction with the main `pea` commands, and not the table or figure specific commands. For the table or figure specific commands, those options in Table 3 need to be specified that relate to the specific table or figure. Which options are required for which table or figure is specified in the help files (accessible through `help pea core`, `help pea tables`, `help pea figures`, `help pea table10`, `help pea figure2`, etc.).⁹ Not specifying `setting(GMD)` can be useful when users want to use categories and values for certain variables that differ from the harmonized default values listed in Table 1.

Every output, both figures and tables, are saved in an Excel sheet, along with table and figure notes. When the `excel()` option is not specified, a temporary Excel file will be created, which is automatically opened after the code is completed. Users can specify where the file should be saved with the `excel()` option. However, the in the option specified Excel file needs to already exist (it can be empty) prior to running the `pea` commands.

1. `pea core`

This first code (in Stata: `pea core`) generates a standardized data annex which is expected to be included in all PEAs. The annex includes main poverty and shared prosperity indicators, as well as multidimensional and sub-group (e.g. by age or education) poverty rates and poverty profiles. The code also benchmarks core statistics for a selected group of peer countries, the country's region and income group averages. Peer countries can be specified by the user and a [benchmarking tool](#) is available to support the selection of countries. Growth incidence curves and the Datt-Ravallion decomposition complement the core outputs.¹⁰

```
pea core [weight] [if exp] [in exp] [,Year(varname numeric)
NATWelfare(varname numeric) NATPovlines(varlist numeric)
PPPWelfare(varname numeric) PPPPovlines(varlist numeric)
PPPyyear(integer) svy std(string) SETting(string) BYInd(varlist
numeric) ONEline(varname numeric) ONEWelfare(varname numeric) missing
latest within3 BENCHmark(string) spells(string) excel(string)
save(string)]
```

2. `pea tables`

The second set of outputs is produced using the `pea tables` command. These expand on the core outputs, providing more detailed statistics on poverty and shared prosperity over the last years, the extent of vulnerability to poverty, poverty profiles as well as drivers of welfare changes. The latter examines how inclusive economic growth has been to those in the bottom of the income distribution, as well as decompositions of welfare changes into within and between group shifts.

⁹ Refer to Annex 1: Full list of tables and figures, for a list of all help files and table and figure numbers.

¹⁰ See [benchmarking tool](#) and [guidance note](#) to identify benchmarking countries.

```
pea tables [weight] [if exp] [in exp] [, Country(string) Year(varname
numeric) natwelfare(varname numeric) natpovlines(varlist numeric)
pppwelfare(varname numeric) ppppovlines(varlist numeric)
PPPyyear(integer) svy std(string) oneline(varname numeric)
onewelfare(varname numeric) setting(string) byind(varlist numeric)
latest within3 earname(integer) missing benchmark(string)
spells(string) excel(string) save(string)]
```

3. pea figures

The third output includes a series of figures that are produced using the `pea figures` command. Figures include a comparison of poverty rates and GDP per capita with other countries, changes in poverty and inequality over time, population composition by income decile, profiles, and exposure to risk from climate-related hazards.¹¹ Options that are specific to the `pea figures` command are listed in Table 4. Each Excel sheet containing a figure also includes the raw data that is used to create the figure (in row 100 of the sheet).

```
pea figures [if] [in] [weight], [natwelfare(varname)
natpovlines(varlist) pppwelfare(varname) ppppovlines(varlist)
PPPyyear(integer) year(varname) setting(string) byind(varlist)
oneline(varname) onewelfare(varname) missing country(string)
within(integer) combine comparability(varname) benchmark(string)
spells(string) noequalspacing scheme(string) palette(string)
welfaretype(string) yrange() ineqind() excel(string) save(string)]
```

¹¹See the [Corporate scorecard website](#) and [CCG dashboard](#) on the risk from climate-related hazards indicator.

Table 2: Main options

Setting	Description	Entry or variable name
1) Required options		
Country()	PEA country	<i>3-letter country code</i>
natwelfare()	Welfare aggregate in current or constant LCU	welfare
natpovlines())	National poverty line, multiple entries allowed	natline
pppwelfare())	Welfare aggregate in PPP terms	welfareppp
ppppovlines())	International poverty lines in PPP terms	pline215 pline365 pline685
pppyear()	specifies which year PPPs are based on. Default is 2017	<i>Integer</i> (such as 2017)
svy	Triggers the inclusion of standard-errors in Table 1 and Table A.1. Importantly, the data needs to be 'svy set' in Stata before the code is run. See Annex 2: Example code (step 3).	
std(string)	Specifies whether standard-errors are displayed in the same cell as the main statistic or in a separate cell to the right. Default is inside.	inside; right
year()	Year variable in survey dataset	year
onewelfare()	Main welfare variable used for comparisons (i.e. Table 1, Table 7, Table 14, Table A1, Table A2, Table A4, Table A4, Figure A1)	welfareppp
oneline()	Main poverty line used for comparisons (i.e. Table 1, Table 7, Table 14, Table A1, Table A2, Table A4, Table A4, Figure A1)	pline685
2) Optional options		
byind()	Geographic units for disaggregation, multiple entries allowed. Note that while optional, it is used in many tables.	urban subnatvar
urban()	Variable that indicates the urban indicator	urban
benchmark()	Countries to be benchmarked against, multiple entries allowed	<i>3-letter country codes</i>
setting()	If GMD option is specified, harmonized variables are created, and additional options (Table 3) in do not need to be specified.	<i>GMD</i>
spells()	Years to be used when calculating or showing changes across periods (such as Growth Incidence Curves), multiple entries allowed separated by semicolon	<i>year1 year2; year2 year3</i>
missing	Missing values are reported separately. Otherwise, they are treated as non-existent.	
minobs()	Specifies the minimum number of observations required to display a cell value (relevant for Tables A2, 2 and 3)	<i>integer</i>
vulnerability())	Specifies the number that is multiplied with the main poverty line to define vulnerability to poverty (relevant for Table 7). Default is 1.5.	<i>Any number</i>

earnage()	Specifies the age cut-off for working status for the economic composition (Table 14).	<i>integer</i>
latest	Includes only the most recent available data.	
Within3	Limits analysis to data from benchmark countries within 3 years of the target year.	
excel()	Path of directory where and name under which output file is saved. If it is not specified, the file is saved in “Temp”. Note that if the option is specified, the excel file needs to exist before the code is run (empty).	<i>Path and file name</i>
save	Specifies the file path for saving results.	

Additional options, if `setting(GMD)` is not specified. Note that these are relevant for the `pea core`, `pea tables` and `pea figures`, and all individual tables and figures commands.

```
hhhead(varname numeric) edu(varname numeric) married(varname numeric)
school(varname numeric) services(varlist numeric) assets(varlist
numeric) hhsize(varname numeric) hhid(string) pid(string)
industrycat4(varname numeric) lstatus(varname numeric) empstat(varname
numeric) relationharm(varname numeric)
```

Table 3: Additional options if `setting(GMD)` is not specified

Setting	Description	Default under <code>setting(GMD)</code>
age()	Age of individual	age
male()	Sex of individual	male
married()	Marital status (variable is based on marital)	married
services()	Services to be included in the analysis. Default are access to improved water, access to improved sanitation, access to electricity.	imp_wat_rec imp_san_rec electricity
assets()	Assets to be included in the analysis	tv car cellphone computer fridge
hhsize()	Household size	hsize
relationharm()	Relation to household head	relationharm
head()	Household head indicator (Variable is based on variable relationharm)	head
hhid()	Household identifier	hhid
pid()	Individual identifier	pid
edu()	Education level	educat4
school()	Currently in school dummy	school
lstatus()	Dummy if respondent does not work (Variable is based on variable lstatus). Please make sure the value = 1 refers to the respondent not working.	nowork

industryat4()	Industry categories	industrycat4
empstat	Employment status of respondent	empstat

Table 4: Additional optional figure options

Setting	Description	Entry or variable name
within()	Specifies the number of years before and after the pea survey year, to define which surveys from other countries should be used (e.g. in scatter plots on inequality). Default is 3, and value should be less than 10.	<i>integer</i>
combine	When specified, figures with multiple panels are combined to one figure with only one legend.	
comparability()	This variable denotes which survey rounds are comparable over time. Non-comparable survey rounds are not connected in figures.	<i>comparability</i>
noequalspacing	When specified, figures show gaps between years on the x-axis proportional to their distance. Default is to display constant gaps between years, regardless of how far away years are. The option can be useful if gaps between survey-years are large.	
scheme()	Sets the scheme, specifying the overall look of the figures. Default is “white_tableau”.	<i>string</i>
palette()	Sets the color palette for figures. Default is “tab20”. See Annex 2: Example code for an example of a custom set of colors.	Either <i>string</i> (e.g. <i>vividis</i>) or list of colors.
welfaretype()	Can be used to specify whether the survey uses income or consumption to calculate welfare. Figures showing scatters of inequality display different symbols for countries with consumption or income aggregates.	<i>CONS</i> or <i>INC</i>
yrange()	Users can specify the range of the y-axis. The range must be entered in Stata figure format.	Example: “0(10)100”
Ineqind()	Users can specify which inequality indicators to show in Figure 9a. Entry options are Gini Theil Palma Top20. Default are all four.	Any combination of: Gini Theil Palma Top20
bar	Users can specify this option to display Figures 1, 9a and 10a as bar graphs instead of line graphs. Warning: All selected years will be shown in the figures, regardless of whether they are comparable or not.	
trim()	Specifies percentiles below and above which growth incidence curves are trimmed.	Example “trim(3 97)”

III. Caveats and FAQ

- Importantly, when the nominal welfare aggregate is passed into the `onewelfare()` option, growth incidence curves will show growth in welfare in nominal, not in real terms. In this case, it is advised to redo the growth incidence curves separately (e.g. `pea table11, welfare(welfppp) spells(2018 2022) year(year) setting(GMD) graph`).
- The notes of Tables 3b, 3c, 14a A4a, and Figures 7a, 7b, specify that education levels refer to “highest level attended, complete or incomplete”. If this does not correspond to the data used in the analysis, please change the notes of these outputs accordingly.
- When there are multiple survey periods, make sure that welfare in LCU and national poverty lines are both expressed in current prices or both in constant prices.
- Note that when national welfare in current prices is specified in the `oneline()` option, growth incidence curves will not reflect changes in real-terms. Therefore, specifying welfare in constant terms in the `oneline()` is encouraged, or re-doing the growth incidence curves separately in constant terms (`pea table11, pea figure3`) is recommended.
- It is important that the correct temporal and spatial deflators are used.
- Poverty numbers are based on population numbers from the respective surveys. Note that this may cause slight discrepancies between poverty numbers reported in PIP and in PEBs, which are based on population from WDI.
- Labor force related variables can have different recall periods. For example, in GMD, some survey data use the variable `lstatus` (which has a 7-day recall period) and some use `lstatus_year` (which has a 1-year recall period). It is up to country teams to decide whether variables are comparable over time (e.g. even when recall periods are different). For the example of labor force status, if different recall periods are deemed to be comparable, one variable would need to be defined which has both variables’ values (e.g. `replace lstatus = lstatus_year if lstatus==.`)
- Furthermore, labor force related variables (`lstatus` and `empstat`) often have many missing values. These variables affect Table A.4a, Table A.4b, Table 13b, Table 14a, Table 14b, Figure 5b and Figure 16. It is advised to inspect the data before running the commands for these tables.
 - T Please make sure the value = 1 of the `lstatus` variable refers to the respondent not working.
- For Table 13b and Figure 5b, which apply Huppi-Ravallion decompositions on the sectoral (agriculture vs. non-agriculture) level, the program assumes that the specified sector variable (default is `industrycat4`), takes value of 1 for the agriculture sector. Please ensure that this is the case in your data.
- The MPM is retrieved from the GMI. New surveys in the GMD might not be immediately included in the GMI (updated usually in April and October).
- For growth incidence curves disaggregated by urban and rural areas, the calculation is performed for the welfare distribution within that area only.
- The “Other” sector in the variable `industrycat4` is based on a country-specific harmonization from the GMD which can vary across countries.
- Note that not all surveys of a country in the GMD are comparable over time, due to updated methodologies. The variable “comparability” indicates which surveys are comparable. When

specified, the `comparability` option in the `pea figures` commands shows breaks between non-comparable survey spells.

FAQ

Q1: Why are my tables missing some indicators?

- ✓ Check if all required variables (Step 3) are present in the dataset.
- ✓ Ensure numerical variables are correctly formatted (encode string variables if needed).

Q2: Can I use country-specific variables instead of GMD harmonized variables for education, sector, etc?

- ✓ Yes, not specifying the option `setting (GMD)` allows to insert country-specific variables in the options specified in Table 3. For example, the globally harmonized education variable in the GMD database takes on only 4 values (`educat4`), but a different, country-specific variable can be inserted in option `edu()`.¹² See example 2b under Step 4 in Annex 2: Example code.

Q3: How do I define a custom color palette for figures?

- ✓ Use the `palette()` option, see example under Step 4 in Annex 2: Example code.

Q4: How do I include multiple years for analysis?

- ✓ Make sure that all variables are consistently specified over all years
- ✓ Options `years()` and `spells()` specify which years are used in the analysis. Use `spells(year1 year2; year3 year4)`, to select which pairs of years should be used to display growth rates (for example in growth incidence curves).

Q5: I get an error that `datalibweb` or `pip` cannot be accessed, what can I do?

- ✓ Make sure that your `datalibweb` token is active (see Step 2a, footnote 10). It needs to be renewed monthly.

¹² In the GMD, education by default follows the International Standard Classification of Education (ISCED) mappings.

✓ PIP or datalibweb could be temporarily not accessible. You can test by running 'pip, clear' or 'dlw' in Stata.

Annex 1: Full list of tables and figures

Standardized data annex (pea core):

- Table A.1. Core poverty and equity indicators
 - [help pea table1](#)
- Table A.2. Poverty indicators by subgroup
 - [help pea tableA2](#)
- Table A.3. Benchmarking of poverty and inequality
 - [help pea table10](#)
- Table A.4a. Profiles of the poor
 - [help pea table4](#)
- Table A.4b. Demographic and economic household typologies
 - [help pea table4](#)
- Figure A.1 Growth Incidence Curves
 - [help pea figure3b](#)
- Figure A.2 Datt-Ravallion decomposition
 - [help pea figure4](#)

Full set of tables (pea tables):

- Table 1. Core poverty indicators
 - [help pea table1](#)
- Table 2. Core poverty indicators by geographic areas
 - [help pea table2](#)
- Table 3a. Subgroup poverty rates by sex and age-group
 - [help pea table3](#)
- Table 3b. Subgroup poverty rates by education (age 16+, %)
 - [help pea table3](#)
- Table 3c. Subgroup poverty rates of household head (%)
 - [help pea table3](#)
- Table 4a. Demographic profiles of the poor
 - [help pea table4](#)
- Table 4b. Labor force profiles of the poor
 - [help pea table4](#)
- Table 5. Key labor market indicators
 - [help pea table5](#)
- Table 6a. Multidimensional poverty: Multidimensional Poverty Measure (World Bank) (%)
 - [help pea table6](#)
- Table 6b. Multidimensional poverty: Multidimensional poverty components (%) (World Bank)
 - [help pea table6](#)
- Table 7. Poverty vulnerability
 - [help pea table7](#)
- Table 8. Core inequality indicators
 - [help pea table8](#)

- Table 9. Scorecard Vision Indicators
 - [help pea table9](#)
- Table 10. Benchmarking of poverty and inequality
 - [help pea table10](#)
- Table 12a. Decomposition of poverty changes: growth and redistribution - Datt-Ravallion decomposition
 - [help pea table12](#)
- Table 12b. Decomposition of poverty changes: growth and redistribution - Shorrocks-Kolenikov decomposition
 - [help pea table12](#)
- Table 13a. Huppi-Ravallion decomposition (rural/urban)
 - [help pea table13](#)
- Table 13b. Huppi-Ravallion decomposition (sectoral)
 - [help pea table13](#)
- Table 14. Demographic and economic household typologies
 - [help pea table14](#)
- Table 15. Distribution of welfare by deciles
 - [help pea table15](#)

Full set of figures (pea figures):

- Figure 1: Poverty rates by year lines.
 - [help pea figure1](#)
- Figure 2. Poverty and GDP per capita in benchmark countries
 - [help pea figure2](#)
- Figure 3a. Growth Incidence Curves over time
 - [help pea figure3a](#)
- Figure 3b. Growth Incidence Curves by area
 - [help pea figure3b](#)
- Figure 4a. Datt-Ravallion decompositions
 - [help pea figure4](#)
- Figure 4b. Shorrocks-Kolenikov decompositions
 - [help pea figure4](#)
- Figure 5a. Huppi-Ravallion decomposition (rural/urban)
 - [help pea figure5](#)
- Figure 5b. Huppi-Ravallion decomposition (sectoral)
 - [help pea figure5](#)
- Figure 6. GDP - Poverty elasticity
 - [help pea figure6](#)
- Figure 7a. Share of poor and population by demographic groups
 - [help pea figure7a](#)
- Figure 7b. Poverty rates by demographic groups
 - [help pea figure7b](#)
- Figure 8. Poverty rates by sex and age groups
 - [help pea figure8](#)

- Figure 9a. Inequality by year
 - [help pea figure9a](#)
- Figure 9b. Gini and GDP per capita scatter
 - [help pea figure9b](#)
- Figure 9c. Benchmark countries ranked by Gini index
 - [help pea figure9c](#)
- Figure 9d. Welfare percentiles by year
 - [help pea figure9d](#)
- Figure 10a. Prosperity gap by year and area
 - [help pea figure10a](#)
- Figure 10b: Prosperity gap (line-up) and GDP per capita scatter
 - [help pea figure10b](#)
- Figure 10c: Prosperity gap (survey) and GDP per capita scatter
 - [help pea figure10c](#)
- Figure 10d: Prosperity gap over time in benchmark countries (line-up)
 - [help pea figure10d](#)
- Figure 12. Decomposition of growth in prosperity gap
 - [help pea figure12](#)
- Figure 13. Distribution of welfare by deciles
 - [help pea figure13](#)
- Figure 14a. Multidimensional Poverty Measure, by components, versus benchmark countries
 - [help pea figure14](#)
- Figure 14b. Multidimensional poverty and poverty rates, versus benchmark countries
 - [help pea figure14](#)
- Figure 14c. Multidimensional poverty and poverty rates, contributions
 - [help pea figure14](#)
- Figure 15: Climate risk and vulnerability
 - [help pea figure15](#)

Annex 2: Example code

The following code produces the main tables, the standardized data annex and figure outputs for Guinea-Bissau.

```
// Set path where the survey data is stored, or should be stored if
accessing from datalibweb
global pea_path "C:/Users/data/ "

// Step 2: Access data (either from survey, or from datalibweb)
datalibweb, country(GNB) year(2018 2021) type(gmd) mod(all) clear

// Step 3: Preparation of additional variables
// Welfare aggregate to compute international poverty rates
gen welfppp = welfare/cpi2017/icp2017/365
gen pline215 = 2.15
gen pline365 = 3.65
gen pline685 = 6.85
// National poverty line: Adjust according to the PEA country
gen natline = 298083.5 if year == 2021
replace natline = 271071.8 if year == 2018
la var pline215 "$2.15 per day (2017 PPP)"
la var pline365 "$3.65 per day (2017 PPP)"
la var pline685 "$6.85 per day (2017 PPP)"
la var natline "National poverty line (2017 LCU)"
// Cleaning of region variable: Adjust according to the PEA country
split subnatid, parse("-") gen(tmp)
replace tmp2 = ustrlower( ustrregexra( ustrnormalize( tmp2, "nfd" ) ,
"\p{Mark}", "" ) )
replace tmp2 = " bolama/bijagos" if tmp2 == " bolama_bijagos"
replace tmp2 = proper(tmp2)
encode tmp2, gen(subnatvar)
// Additional labeling for table and figure notes
label var welfppp "Welfare (constant 2017 PPP)"
label var welfare "Welfare (local currency)"
local lbl: value label educat4
label define `lbl' 1 "No education" 2 "Primary" 3 "Secondary" 4
"Tertiary", modify
// Survey set the data, to get correct standard errors
svyset psu [w= weight_p], singleunit(certainty)

save "${pea_path}/data/GNB_GMD_clean.dta", replace

// Step 4: Run pea commands
// 1. Main tables code:
//Set ado path
adopath + "${pea_path}/ado/pea/Stata/plus"
```

```

//Call data
use "${pea_path}/data/GNB_GMD_clean", clear
//Run code with options
pea tables [aw=weight_p], c(GNB) year(year) ///
    natw(welfare) natp(natline) ///
    pppw(welfppp) pppp(pline365 pline215 pline685) ///
    onew(welfppp) oneline(pline365) ///
    byind(urban subnatvar) vulnerability(2) ///
    benchmark(CIV GHA GMB SEN) ///
    setting(GMD) missing ///
    spells(2018 2021) ///
    svy std(inside)

// 2a. Appendix tables code:
//Set ado path
adopath + "${pea_path}/ado/pea/Stata/plus"
//Call data
use "${pea_path}/data/GNB_GMD_clean", clear
//Run code with options

pea core [aw=weight_p], c(GNB) year(year) ///
    natw(welfare) natp(natline) ///
    pppw(welfppp) pppp(pline365 pline215 pline685) ///
    onew(welfppp) oneline(pline365) ///
    byind(urban subnatvar) ///
    benchmark(CIV GHA GMB SEN) ///
    setting(GMD) missing ///
    spells(2018 2021) ///
    svy std(inside)

// 2b. Appendix tables code without setting(GMD):
//Set ado path
adopath + "${pea_path}/ado/pea/Stata/plus"
//Call data
use "${pea_path}/data/GNB_GMD_clean", clear
//Run code with options
pea core [aw=weight_p], c(GNB) year(year) ///
    natw(welfare) natp(natline) ///
    pppw(welfppp) pppp(pline365 pline215 pline685) ///
    onew(welfppp) oneline(pline365) ///
    byind(urban subnatvar) ///
    benchmark(CIV GHA GMB SEN) ///
    missing spells(2018 2021) ///
    svy std(inside) ///
    age(age) male(male) hhhead(head) edu(educat4) ///
    urban(urban) married(married) school(school) ///
    services(imp_wat_rec imp_san_rec electricity) ///

```

```

        assets(tv car cellphone computer fridge)          ///
        hhsz(hsize) hhid(hhid) pid(pid)                  ///
        industrycat4(industrycat4) relationharm(relationharm) ///
        lstatus(nowork) empstat(empstat)

// 3a. Figures code:
//Set ado path
adopath + "${pea_path}/ado/pea/Stata/plus"
//Call data
use "${pea_path}/data/GNB_GMD_clean", clear
//Run code with options
pea figures [aw=weight_p], c(GNB) year(year)            ///
        natw(welfare) natp(natline)                      ///
        pppw(welfppp) pppp(pline365 pline215 pline685)   ///
        onew(welfppp) oneline(pline215)                  ///
        byind(urban) benchmark(CIV GHA GMB SEN AGA)      ///
        spells(2010 2018; 2018 2021)                    ///
        setting(GMD) urban(urban) within(3)              ///
        comparability(comparability) welfaretype(CONS)   ///
        combine

// 3b. Figures code but bars instead of lines and non-equal spacing
between years:
//Set ado path
adopath + "${pea_path}/ado/pea/Stata/plus"
//Call data
use "${pea_path}/data/GNB_GMD_clean", clear
//Run code with options
pea figures [aw=weight_p], c(GNB) year(year)            ///
        natw(welfare) natp(natline)                      ///
        pppw(welfppp) pppp(pline365 pline215 pline685)   ///
        onew(welfppp) oneline(pline215)                  ///
        byind(urban) benchmark(CIV GHA GMB SEN AGA)      ///
        spells(2010 2018; 2018 2021)                    ///
        setting(GMD) urban(urban) within(3)              ///
        comparability(comparability) welfaretype(CONS)   ///
        combine
        bar noequalspacing

// 4. Running a single figure and defining own color palette
use "${pea_path}/data/GNB_GMD_clean", clear
adopath + "${pea_path}/ado/pea/Stata/plus"
local custom_palette = "#337ab7 #5cb85c #5bc0de #f0ad4e #d9534f #e6e6e6
#286090 #449d44 #31b0d5 #ec971f #c9302c"

pea figure2 [aw=weight_p], c(GNB) year(year)            ///
        onew(welfppp) onel(pline215)                    ///
        benchmark(CIV GHA GMB SEN) palette(`custom_palette')

```

// 5. Running a single table with LCU poverty line

```
use "${pea_path}/data/GNB_GMD_clean", clear
adopath + "${pea_path}/ado/pea/Stata/plus"

// When running single tables, setting(GMD) is not called, and additional
variables need to be created
gen head = relationharm==1 if relationharm~=.
la def head 1 "HH head"
la val head head
gen nowork = lstatus==2|lstatus==3 if lstatus~=.
label define nowork 0 "Working" 1 "Not working (unemployed or out of
labor force)"
label values nowork nowork
gen married = marital==1 if marital~=.

pea table4 [aw=weight_p], welfare(welfare) povlines(natline)      ///
      year(year) missing urban(urban)                          ///
      age(age) male(male) hhhead(head)                         ///
      edu(educat4) married(married) school(school)             ///
      services(imp_wat_rec imp_san_rec electricity)            ///
      assets(tv car cellphone computer fridge)                 ///
      hhsz(hsize) hhid(hhid) pid(pid)                          ///
      industrycat4(industrycat4) lstatus(nowork)                ///
      empstat(empstat) relationharm(relationharm)
```